

Pràctiques Integrades

1er de Matemàtiques

Pràctica 25

curs 2002–03

25 Càlcul de primitives

El Teorema Fonamental del Càlcul ens diu que tota funció integrable Riemann en un interval $[a, b]$ té una primitiva en aquest interval. Aquesta pràctica està dedicada al càlcul de primitives i es fa un breu repàs de les tècniques més habituals. Recordeu d'una pràctica anterior que si f és una expressió en x aleshores la comanda de Maple per obtenir la integral definida o de Riemann $\int_a^b f(x)dx$ és `int(f(x), x=a..b)`. Per a resoldre la integral indefinida o càlcul d'una primitiva s'utilitza `int(f(x), x)`.

Exemple 25.1

```
> i:=int(1/(1+x^2),x);
```

És fàcil comprovar que el resultat és correcte fent una derivada de la funció obtinguda.

```
> diff(i,x);  
> diff(i+C,x);
```

També coneixem la comanda `Int`. Recordeu que és una comanda inert que només reproduïx per pantalla el que li hem demanat (de manera que podem comprovar que no ens hem equivocat a l'introduir la funció) i que per a obtenir el càlcul explícit s'ha de combinar amb la comanda `value`.

Exemple 25.2

```
> Int(1/(1+x^2),x);  
> value(%);
```

Ara bé, no tota funció primitiva es pot descriure en termes de *funcions elementals*. Fixeu-vos que passa quan li demaneu a Maple que calculi la funció primitiva de e^{-x^2} .

Exemple 25.3

```
> Int(exp(-x^2),x); value(%);
```

Observeu que en el resultat surt una nova funció que Maple anomena **erf**. Utilitzeu l'ajuda de Maple per a trobar la definició de la funció **erf(x)** i calculeu la seva derivada.

```
> ?erf;
> diff(erf(x),x);
```

De les funcions primitives que no es poden expressar com a combinació de funcions elementals se'n diu funcions el·líptiques. Maple els hi assigna un nom com podeu veure a l'exemple anterior. En el següent exemple veureu unes quantes primitives més d'aquest tipus.

Exemple 25.4

```
> int(sqrt(x^3+1),x);
> int(1/(1-3*x^2)/sqrt(1-x^2)/sqrt(1-4*x^2),x);
> int(sqrt(1-4*x^2)/sqrt(1-x^2),x);
```

Amb l'ajuda de Maple podeu veure els diferents tipus de funcions el·líptiques que té definides. Tot seguit anem a veure un altre tipus de situació en què ens podem trobar.

Exemple 25.5

```
> g:=Int((3*x^2+1)/sqrt((1-x-x^3)*(1+x+x^3)),x);
> value(g);
```

Malgrat que Maple no pot avaluar la integral anterior directament amb la comanda **int** disposa de les tècniques estàndard d'integració en el paquet **student** per ajudar-nos (recordeu que abans d'utilitzar les comandes del paquet heu de fer **with(student):**). Amb aquestes eines es pot simplificar l'expressió de la integral indefinida de manera que Maple pot finalment calcular-la. Analitzarem els mètodes de canvi de variable, integració per parts i estudiarem el cas en que la funció és un quocient de polinomis.

25.1 Canvi de variable o substitució

La integració per substitució o canvi de variable es basa en la regla de la cadena i és efectiva si tenim una integral expressable en la forma $\int f(g(x))g'(x)dx$. Aleshores, si fem la substitució $g(x) = u$ i per tant $g'(x)dx = du$, resultarà una nova integral indefinida de la forma $\int f(u)du$ que, si hem fet el canvi adequat, serà més fàcil de calcular.

Maple aplica un canvi de variable en una integral amb la comanda **changevar**. La sintaxi d'aquesta comanda és **changevar(f(u)=h(x),integral,u)** on **integral** és la integral en la variable **x**, **f(u)=h(x)** és la substitució que desitgem fer i **u** és la nova variable en la integral.

Exemple 25.6

```
> f:=Int((exp(x)-2*exp(2*x))/(1-exp(x)),x);
> g:=changevar(exp(x)=t,f,t);
> h:=simplify(%);
> q:=value(h);
```

Ara bé, ens queda la tasca de desfer el canvi de variable fet: escriure la solució en termes de la variable original. Per això només cal fer una substitució (**subs**).

```
> F:=subs(t=exp(x),q)+C;
```

Exercici 25.1

Quin és el canvi més adient per a resoldre el problema de l'exemple 25.5?

Exercici 25.2

Simplifiqueu al màxim mitjançant un canvi de variable el càlcul de primitives per les següents funcions:

a) $\int \frac{x+1+2\sqrt{x}}{x-\sqrt{x}} dx$

b) $\int \frac{e^x - 2e^{-x}}{1 - e^x} dx$

c) $\int \cos(x)^5 dx$ (recordeu la relació trigonomètrica $\sin(x)^2 + \cos(x)^2 = 1$).

25.2 Mètode d'integració per parts

El mètode d'integració per parts es desprèn de la regla de la derivació del producte de funcions. Usualment, s'escriu $\int u dv = uv - \int v du$. Una vegada optem per aquest mètode, el problema bàsic és decidir quina funció cal prendre com a u .

La comanda Maple per integrar per parts és `intparts(integral,u)`, on u indica l'expressió que correspon a la part u en la descomposició $u dv$ abans indicada. Maple ens informa sobre els detalls de la integració per parts en cada problema. No calcula la primitiva sinó que aplica la fórmula amb les dades que li passem.

Exemple 25.7

```
> f:=Int(x*exp(2*x),x);
> intparts(f,x);
> value(%);
```

Exemple 25.8

Anem a calcular una primitiva per $(3x^2 - 2x - 7) \cos(x)$ utilitzant el mètode d'integració per parts. Prenem $u = 3x^2 - 2x - 7$.

```
> i:=Int((3*x^2-2*x-7)*cos(x),x);
> i1:=intparts(i,3*x^2-2*x-7);
```

Per extreure la integral en el segon terme, utilitzem la comanda `op` on indiquem primer l'ordre de la part que volem aïllar.

```
> r1:=op(1,i1);i2:=op(2,i1);
```

i podem procedir amb una nova integració per parts si cal.

```
> intparts(i2,6*x-2);
```

```
> r2:=value(%);
```

```
> resultat:=r1+r2;
```

Què passa si prenem $u = \cos(x)$?

Exercici 25.3

Amb un canvi de variable o bé integrant per parts les vegades que sigui necessari, simplifiqueu i calculeu les següents integrals indefinides:

a) $\int \frac{dx}{\cos(x)}$

b) $\int x^n \ln(x) dx$

c) $\int x^n e^x dx$

d) $\int x \arctan(x) dx$

e) $\int \cos(x) e^x dx$

Exercici 25.4

Calculeu $\int x^2 \arctan(x) dx$. Si feu $u = \arctan(x)$ haureu de resoldre una integral d'una funció racional, problema que comentarem a la secció següent.

25.3 Integrals racionals

Una funció racional és una funció que s'expressa com un quocient de polinomis. En aquest apartat analitzarem les diferents primitives que ens podem trobar a partir de funcions racionals així com el mètode per a calcular-les.

Abans de tot però anem a veure com, donada una expressió racional, tenim comandes que ens aïllen els polinomis del numerador i del denominador (**numer** i **denom**).

Exemple 25.9

```
> rac:=(x^2+3*x-6)/(x^5-7);
> p1:=numer(rac);
> p2:=denom(rac);
```

Anem a analitzar uns casos senzills de funcions racionals.

Exercici 25.5

- a) Fent un canvi de variable calculeu una primitiva per la funció $\frac{1}{3x-5}$. Quin és el canvi de variable més adient per calcular una integral del tipus $\frac{1}{ax+b}$?
- b) Calculeu una primitiva per $\frac{1}{(x-5)^2+9}$. En què es transforma si fem el canvi de variable $x-5 = 3 \tan(u)$? Calculeu la primitiva, quina és la substitució que cal fer per a desfer el canvi?

El següent cas que analitzarem és el cas $\frac{1}{(ax^2+bx+c)^n}$. Aquest tipus d'integrals se simplifiquen si es reescriu l'expressió ax^2+bx+c de manera que sigui del tipus $a(x-d)^2+e$. D'aquesta tècnica se'n diu "completar quadrats". És fàcil comprovar que $d = \frac{-b}{2a}$ i $e = \frac{-b^2}{4a} + c$. El paquet **student** conté la comanda **completesquare** que fa aquesta tasca.

```
> completesquare(2*x^2-6*x+18,x);
```

Aleshores el canvi que cal fer (com ja hem vist a l'exercici anterior) és $x-d = \sqrt{e} \tan(u)$ i després utilitzar la relació trigonomètrica $1 + \tan(u)^2 = \frac{1}{\cos(u)^2}$ si cal.

Exemple 25.10

```
> r:=1/(x^2-6*x+18)^3;
> d:=denom(r);
> d1:=op(1,d);
> d2:=completesquare(d1,x);
> r1:=Int(r,x);
> r2:=subs(d1=d2,r1);
> r3:=changevar(x-3=3*tan(u),r2,u);
> r4:=simplify(r3);
```

Hem de forçar Maple a fer la darrera substitució $1 + \tan(u)^2 = \frac{1}{\cos(u)^2}$.

```
> r5:=subs(1+tan(u)^2=1/cos(u)^2,r4);
> r6:=simplify(r5);
> r7:=value(%);
> resultat:=subs(u=arctan((x-3)/3),r7); simplify(%);
```

Exercici 25.6

Calculeu una primitiva de la funció $\frac{3}{(2x^2 - 8x + 36)^2}$ amb el mètode anterior.

Anem a veure tot seguit què passa si al cas anterior afegim un polinomi de grau 1 al numerador.

Exercici 25.7

Calculeu una primitiva per la funció $\frac{2x+2}{x^2+2x+11}$ mitjançant un canvi de variable. Fixeu-vos que estem davant un cas particular de la situació general $\frac{2ax+b}{ax^2+bx+c}$.

Ara bé, fixeu-vos que donada una funció racional del tipus $\frac{dx+e}{ax^2+bx+c}$ es pot descompondre com a suma de dues dels tipus que ja hem estudiat:

$$\frac{\frac{c}{2a}(2ax+b)}{ax^2+bx+c} + \frac{d - \frac{eb}{2a}}{ax^2+bx+c}.$$

Finalment, observeu que tots els casos vistos fins ara corresponen a tipus de funcions racionals en que el grau del numerador és menor que el grau del denominador. En general i donada qualsevol funció racional sempre ens podem reduir a aquesta situació utilitzant la divisió de polinomis. Si $\frac{P}{Q}$ és una funció racional, la podem escriure com $\frac{P}{Q} = q + \frac{r}{Q}$ on q i r son el quocient i el reste de la divisió de polinomis. Aleshores el grau de r és menor que el grau de Q . Recordeu les comandes **quo** i **rem** per calcular el quocient i el reste de la divisió.

```
> rac1:=(4*x^3+8*x-7)/(x^2-6);
> num:=numer(rac1);
> den:=denom(rac1);
> q:=quo(num,den,x);
> r:=rem(num,den,x);
> rac2:=q+r/den;
```

Per acabar, tota funció racional en la que el grau del numerador és menor que el del denominador es pot escriure com a suma de funcions dels tipus estudiats al principi de la secció. D'aquesta descomposició se'n diu la descomposició en fraccions parcials. Maple calcula aquesta descomposició directament mitjançant **convert**.

Exemple 25.11

```
> convert(rac1,parfrac,x);
> w:=(x^4+2)/(x^3+x^2+x);
> convert(w,parfrac,x);
```

Observeu que el càlcul d'una primitiva per a w és immediat després de expressar-la com a suma de fraccions del tipus estudiat.

Exercici 25.8

Donades $f(x) = \frac{3x - 7}{(x - 1)^2(x - 2)^3}$ i $g(x) = \frac{x^5 + x + 2}{x^3 + x^2 + x}$.

- Doneu la descomposició de f i g en suma de fraccions del tipus estudiat al principi de la secció.
- Per a cadascuna de les funcions racionals obtingudes a l'apartat anterior, digueu quin és el canvi (o canvis) necessaris per a resoldre-la.
- Calculeu una primitiva de f i una de g .